Процедуры и функции

1. Простейшая процедура

Часто одну и ту же последовательность инструкций требуется повторить в нескольких местах программы. Чтобы не тратить время на копирование инструкций в большинстве языков программирования предусмотрены средства для организации подпрограмм. Подпрограмма – последовательность инструкций, которой можно дать произвольное имя и использовать его в качестве сокращенной записи. Такую именованную последовательность инструкций будем называть также процедурой. Определение сокращенной записи называется описанием процедуры, а вызов ее из программы вызовом процедуры или оператором процедуры.
Пример 1: Создадим программу с процедурой, печатающей на экране слово Hello.

program HelloProc;

<Раздел описания переменных, типов и т.д.>

procedure P1; 
{Заголовок процедуры. P – имя процедуры}

begin 
{Начало тела процедуры}


writeln(‘Hello’);

end; 
{Конец тела процедуры}

begin 
{Начало программы}


P1; 
{Вызов процедуры}
<Еще какие-то инструкции>

P1;
{Еще один вызов процедуры}

end. 
{Конец программы}

В этом примере мы видим следующее:

1) Описание процедуры располагается в разделе описаний программы (там же где описываются переменные, типы, константы и т.п.). Традиционно описания процедур ставятся в конец раздела (после того как описаны все переменные), но это не обязательно.

2) Простейшая процедура состоит из заголовка и тела процедуры. Заголовок состоит из слова procedure и имени процедуры. Имена процедур (как и имена переменных, типов и констант) могут быть любым сочетанием латинских букв, цифр и символа подчеркивания, начинающимся не с цифры.

3) Тело процедуры ограничено словами begin и end. После end’а ставится точка с запятой. В теле процедуры пишутся все инструкции, которые будут выполняться при ее вызове.

4) Вызов процедуры производится в разделе операторов программы. Для вызова достаточно написать имя процедуры. В приведенном примере вызов производится два раза. Соответственно дважды будет напечатано слово Hello.
2. Локальные переменные
Каждая процедура может иметь собственный раздел описания переменных (а также типов, констант и т.п.). Переменные, описанные в этом разделе, называются локальными и действуют только внутри процедуры.

Часто некоторые переменные используются только внутри последовательности инструкций и не имеют смысла за пределами этих инструкций. В программе существенно проще разобраться, если такие инструкции будут оформлены как процедура, а такие переменные будут описаны как локальные.

Пример 1: Программа с процедурой, печатающая слово Hello 10 раз.

var


i: integer;
procedure Hello10;

var
{Открытие раздела описаний локальных переменных}


i: integer;
{Описание локальной переменной i}

begin

for i:=1 to 10 do write(‘Hello_’);
end;
begin


for i:=1 to 5 do Hello10;
{Пять раз вызываем процедуру}
end.
Программа выведет 5 строк по 10 слов Hello.

Как видим, раздел описаний процедуры располагается сразу после заголовка и перед телом процедуры. В разных процедурах можно использовать локальные переменные с одинаковыми именами. Не смотря на одинаковость имен, это будут разные переменные, их значения будут храниться в разных областях памяти, а присваивания одним никак не повлияют на значения других. После завершения работы процедуры память, выделенная под локальные переменные, освобождается, все их значения пропадают.

Переменные, описанные в разделе описаний основной программы, будем называть глобальными. Глобальные переменные действуют как в теле программы, так и во всех процедурах. Если имя локальной и глобальной переменной, как в приведенном примере, совпадают, то это все равно разные переменные. Внутри процедуры при обращения к переменой i подразумевается, что это локальная переменная.

Пример 3: Глобальная и локальная переменные

var

x: real;
{Описание глобальной переменной x}

procedure P2;

var

x: real;
{Описание локальной переменной x}

begin

x:=10;
{Присваивание значения локальной переменной}

end;

begin

x:=5;
{Присваивание значения глобальной переменной}


P2;
{Вызов процедуры}


writeln(x);
{Вывод значения глобальной переменной}

end.
Не смотря на то, что при вызове процедуры P2 выполнится инструкция x:=10; глобальная переменная x останется равной 5.

Представьте себе, что в предыдущем примере (за номером 3) локальная переменная i не описана. Синтаксической ошибки не возникнет, так как есть глобальная переменная, которая и будет использована как счетчик цикла внутри процедуры. Однако смысл программы измениться. Процедура изменит значение глобальной переменной i, которая используется как счетчик в другом цикле уже в основной программе. Последствия труднопредсказуемы. Скорее всего, программа выведет только одну строку со словами Hello. Но может и зациклится, выводя Hello до бесконечности.

3. Зачем на самом деле нужны процедуры
Вряд ли стоило бы уделять много внимания процедурам, если бы за ними не скрывались важные и основополагающие понятия. В действительности процедуры оказывают решающее влияние на стиль и качество работы программиста. Процедура – это способ сокращения текста, но что более важно, средство разложения программы на логически связанные, замкнутые компоненты, определяющие ее структуру.

Представьте себе программу, содержащую, например, 1000 строк кода (это еще очень маленькая программа). Обозреть такое количество строк и понять, что делает программа, было бы практически невозможно без процедур.

Большие программы строятся методом последовательных уточнений. На первом этапе внимание обращено на глобальные проблемы, и в первом эскизном проекте упускаются из виду многие детали. По мере продвижения процесса создания программы, глобальные задачи разбиваются на некоторое число подзадач. Те в свою очередь на более мелкие подзадачи и т.д., пока решать каждую подзадачу не станет достаточно просто. Такая декомпозиция и одновременная детализация программы называется нисходящим методом программирования.

Концепция процедур позволяет выделить отдельную подзадачу как отдельную подпрограмму. Тогда на каждом этапе можно придумать имена процедур для подзадач, вписывать в раздел описаний их заголовки и, не еще добавляя к ним тело процедуры, уже использовать их вызовы для создания каркаса программы так, будто процедуры уже написаны.

Пример 4: Каркас программы сортировки массивов.

program ArraySort;

var


x: array [1..10] of real;

procedure InputArray;
{Процедура для ввода значений элементов массива}

procedure PrintArray;
{Процедура для вывода элементов массива}

procedure SortArray;
{Процедура сортировки}

begin

InputArray; 
{Вызов процедуры ввода массива}


SortArray;
{Вызов процедуры сортировки}


PrintArray;
{Вызов процедуры вывода}

end.

Какие именно инструкции будут выполняться процедурами InputArray, PrintArray и SortArray пока не сказано. Это следующий этап написания программы. Задачу сортировки в свою очередь можно разбить на более простые подзадачи, а именно многократное повторение поиска минимального элемента и постановка его на нужное место. Эти задачи также можно оформить в виде процедур, создав каркас процедуры SortArray.

Когда каркас создан, остается только написать тела процедур. Преимущество такого подхода в том, что, создавая тело каждой из процедур, можно не думать об остальных процедурах, сосредоточившись на одной подзадаче. Кроме того, когда каждая из процедур имеет понятный смысл, гораздо легче взглянув на программу, понять, что она делает. Это в свою очередь позволит: (а) допускать меньше логических ошибок и (б) организовать совместную работу нескольких программистов над большой программой.

4. Параметры процедуры
Очень часто некоторые последовательности инструкций, встречающиеся в программе, не идентичны, но очень близки по форме. Особенно важна ситуация, когда различие между различными вхождениями инструкций можно устранить систематической заменой идентификаторов или выражений.

Пример: В описанном в лекции алгоритме сортировки, требовалось искать минимальный элемент в несортированной части массива и ставить его на левый край несортированной части. Отличие в действиях на каждом шаге алгоритма сводилось к изменению одного параметра – номера индекса, с которого начиналась несортированная часть.

Такие инструкции можно записать в виде процедуры, работа которой зависит от одного или нескольких параметров. Параметры описываются в скобках после имени процедуры.

Пример 5: Программа, печатающая произведение двух чисел с помощью процедуры.

var

x, y: integer;
procedure Mult(a, b: integer);
{Заголовок процедуры с описанием двух параметров}

var

c: integer;

begin

c:=a*b;
{Значения параметров используются для вычисления c}


writeln(c);

end;

begin


readln(x, y);


Mult(x, y);
{Вызов процедуры с указанием значений параметров}


readln;

end.

Программа запросит у пользователя значения переменных x и y, затем вызовет процедуру Mult, указав при этом, что значения параметров a и b при данном вызове процедуры должны быть равны значениям переменных x и y. Процедура выведет произведение x*y.
Переменные a и b в заголовке процедуры называются формальными параметрами. То, что подставляется при вызове процедуры (в примере это переменные x и y), называется фактическими параметрами.

Формальный параметр внутри процедуры может использоваться как обычная локальная переменная. То есть их можно использовать в выражениях и даже присваивать им новые значения. Отличие от просто локальных переменных в том, что при вызове процедуры обязательно указывается, чему они изначально равны.

В качестве фактических параметров в данном примере могут выступать произвольные выражения. Например, допустим такой вызов процедуры: Mult(x/y+1, 2). Значения этих выражений присвоятся формальным параметрам, и процедура напечатает значение выражения 2(x/y+1).
5. Параметры-значения и параметры-переменные
Существует два способа описывать параметры: как параметры-значения и как параметры-переменные. В приведенном выше примере использовались параметры-значения. Параметр-значение является просто локальной переменной, начальное значение которой задается при вызове процедуры.

Пример 6:

var

a, b: integer;
{Глобальные переменные}
procedure P3(a, b: integer);
{Внутри процедуры символами a и b обозначаются

begin
формальные параметры. Действия с ними никак не повлияют

a:=a+1;
на значения глобальных переменных a и b}


b:=b+1;


writeln(a+b);

end;

begin

a:=1;


b:=1;


P3(a, b);
{Вызов процедуры, где в качестве фактических параметров 


использованы значения глобальных переменных a и b}

writeln(a, b);

end.

Если в заголовке процедуры перед какими-либо из параметров поставить слово var, то это будут параметры переменные. Например:

procedure P3(var a, b: integer; c: real);

Здесь a и b – параметры-переменные, c – параметр-значение. При вызове процедуры фактический параметр, задающий значения этих формальных параметров не может быть выражением или константой, а должен быть обязательно переменной. То есть, недопустимы следующие вызовы:

P3(x+y, y+1, z+2);

P3(x, 2, z);

где x, y, z – глобальные переменные. Такие инструкции вызовут синтаксическую ошибку. В то же время синтаксически правильными будут вызовы:

P3(x, y, z+1);

P3(x, y, 2);

При использовании параметров-значений во время вызова процедуры происходит присваивание значения формальному параметру, то есть значение записывается в выделенные под хранение параметра ячейки памяти. При использовании параметров-переменных память под них не выделяется. Вместо этого на время работы процедуры они становятся синонимами тех переменных, которые указываются в качестве фактических параметров. Так при вызове P3(x, y, 2) формальные параметры a и b будут работать с теми же ячейками памяти, что и переменные x, y. Соответственно, если формальному параметру в процедуре присваивается новое значение, то и значение фактического параметра изменится. Воспроизведем пример 6, заменив тип параметров (добавим var перед их описанием):

var

a, b: integer;
{Глобальные переменные}
procedure P3(var a, b: integer);
{Внутри процедуры символами a и b обозначаются

begin
формальные параметры. Теперь это параметры-переменные и 


a:=a+1;
действия с ними никак повлияют на значения глобальных 


b:=b+1;
переменных a и b}

writeln(a+b);

end;

begin

a:=1;


b:=1;


P3(a, b);
{Вызов процедуры, где в качестве фактических параметров 

использованы значения глобальных переменных a и b. 


После вызова глобальные переменные увеличатся на 1}

writeln(a, b);

end.

Если без слова var программа выводила числа 4, 1, 1, то после его добавления получится 4, 2, 2.
6. Передача массивов в качестве параметров
Как вы знаете с отдельными элементами массивов можно работать также как с обычными (скалярными) переменными. Соответственно можно использовать их в качестве фактических параметров. Однако, может потребоваться передать процедуре не отдельные элементы, а весь массив целиком. При этом не допустим такой, например, способ введения параметра:

procedure P4(a: array [1..10] of integer);

Тип параметра должен быть или скалярным или заранее определенным в разделе type. То есть правильным будет следующее описание:

type

Massive = array [1..10] of integer;

var


x: Massive;

procedure P4(a: Massive);
Такую функцию P можно вызывать, указав в качестве фактического параметра, например, глобальную переменную x.

Пример 7: Программа, вычисляющая сумму элементов массива.

type


Massive = array [1..10] of integer;

var


x: Massive;


i, sum: integer;

procedure P4(a: Massive; var s:integer);
var

i: integer;

begin

s:=0;
{Операции с параметром-переменной s отразятся на

for i:=1 to 10 do
фактическом параметре – глобальной переменной sum}



s:=s+a[i];
end;

begin


for i:=1 to 10 do


readln(x[i]);

P4(x, sum);
{К моменту вызова процедуры, переменной sum ничего еще 



не присвоили. Это не страшно, так как начальное значение 



параметра s несущественно для результата ее работы}


writeln(sum);

end.

При вызове процедуры содержимое массива x будет скопировано в массив a. Глобальная переменная sum на время работы процедуры приобретет синоним s. То есть, присваивая что-то переменной s, мы тем самым изменяем и переменную sum. В итоге программа выведет сумму элементов введенного пользователем массива.
При передаче массива через параметр-значение происходит копирование его содержимого в новые ячейки памяти. При работе с массивами большого размера или при необходимости вызывать процедуру очень много раз это может привести к большим затратам времени. Чтобы этого избежать рекомендуется всегда передавать массивы через параметры-переменные. При этом фактического переноса содержимого массивов из одной области памяти в другую не происходит, и вызов процедуры занимает меньше времени. То есть правильно:

procedure P4(var a: Massive; var s: integer);
7. Открытые массивы
Требование всегда описывать тип-массив для передачи параметра-массива затрудняет разработку универсальных процедур. Например, хотелось бы иметь процедуру, которая может находить сумму элементов произвольного массива, а не только массива из 10 элементов типа integer. Для этого в Паскале предусмотрены так называемые открытые массивы.

Открытым массивом называется формальный параметр-массив, для которого указан тип элементов, но не указана его длина. Например:

procedure P5(var a: array of integer);

В качестве фактического параметра можно передавать целочисленный массив любой длины. Формальный параметр a при этом будет массивом, индексы которого начинаются с нуля. Максимальный индекс в открытом массиве можно получить с помощью стандартной функции High(a). Длину фактически переданного массива можно таким образом подсчитать как High(a) + 1.

Пример 8: Универсальная процедура, подсчитывающая сумму элементов массива.

var


x: array [-5..5] of integer;


i, sum: integer;

procedure ArraySum(var a: array of integer; var s:integer);
var

i: integer;

begin

s:=0;

for i:=0 to High(a) do


s:=s+a[i];
end;

begin


for i:=-5 to 5 do


readln(x[i]);

ArraySum (x, sum);


writeln(sum);

end.

Не смотря на то, что в качестве фактического параметра использован массив со значениями индексов от -5 до 5, формальный параметр есть массив, индексы которого начинаются с нуля. Функция High(a) вернет в данном примере значение 10.
8. Функции
Желаемым результатом работы подпрограммы может быть всего одно значение. Так, например, было в примере с расчетом суммы элементов массива. В этом случае вместо процедур разумно использовать другой вид подпрограмм, а именно функции.

Пример 9: Функция, вычисляющая сумму своих аргументов.

function Sum(a, b: real): real;

begin


Sum:=a+b;

end;

Мы видим, что структура функции почти повторяет структуру процедуры. Отличия:

1) Вместо слова procedure пишется слово function.

2) После списка параметров через двоеточие указывается тип значения, которое получится в результате работы функции.

3) В теле функции должно присутствовать присваивание значения идентификатору функции (Sum:=a+b). Последнее присвоенное значение и будет значением функции.

Вызов функции из программы происходит также, как и в случае со стандартными функциями, таким как sqr, sqrt, random, sin, round и т.д. То есть их значения можно присваивать переменным или использовать их в выражениях. Например, в программе могли бы стоять вызовы:

x:=sum(2, 2);

z:=sum(x, y);

x:=2*sum(x, y)+1;

z:=sqrt(sum(sqr(x), sqr(y)));
z:=sum(sum(x, y), z);
и т.п.

С точки зрения синтаксиса допустим вызов функции без присваивания ее значения какой-нибудь переменной. Например, в программе могла бы быть строчка

sum(2, 2);

и все. Созданное функцией значение в этом случае никуда не запишется. Попросту пропадет. Впрочем, функция может сделать еще что-нибудь полезное. Например, изменить один из своих параметров-переменных или глобальную переменную. Так что такой вызов может иметь смысл.
Правила передачи параметров те же самые, что и для процедур. Также можно описывать локальные переменные.

Для примера приведем функцию, вычисляющую сумму элементов вещественнозначного массива.

Пример 10: Функция, вычисляющая сумму элементов массива.

function ArraySum(var a:array of real): real;

var


s: real;


i: integer;

begin


s:=0;


for i:=0 to High(a) do s:=s+a[i];


ArraySum:=s;

end;

9. Опережающее описание

В теле каждой процедуры или функции может содержаться вызов других процедур или функций при условии, что они описаны раньше, чем процедура или функция их вызывающая. Однако есть возможность вызывать и те процедуры, которые описаны после вызывающей. Для этого надо скопировать заголовок вызываемой подпрограммы и разместить его выше всех описаний. Например:

function F1(x: real): real;
{Опережающее описание функции F1, заголовок без тела функции}

procedure P6;

var


x: real;

begin

<Какие-то действия>

x:=F1(x);
{Вызов функции F1 возможен благодаря опережающему описанию}


<Какие-то действия>

end;
function F1(x: real): real;
{Описание функции F1, теперь тело функции присутствует}

begin

<Какие-то действия>

F1:=<Что-то>

end;

10. Процедурные типы
Значением переменной процедурного типа является целая процедура или функция. Например, в ситуации, когда нужно многократно повторять действия, совпадающие с точностью до замены какого-то выражения или небольшого набора операторов, это выражение или набор операторов можно передать в процедуру в качестве параметра процедурного типа.

Для объявления процедурного типа используется заголовок процедуры или функции без указания имени. Например:

type


TProc1 = procedure (a, b, c: real; var d: real);


TProc2 = procedure (var a, b: array of integer);


TProc3 = procedure;
{Процедура без параметров}


TFunc1 = function: real;
{Функция без параметров}


TFunc2 = function (var x:array of integer): integer;

var


Proc1: TProc1;

Proc2: TProc2;


Proc3: TProc3;

Func1: TFunc1;


Func2: TFunc2;

Если в программе описаны процедуры или функции с подходящим видом заголовка, то их можно присваивать переменным процедурного типа. Например, если есть функция с заголовком

function ArraySum(var a: array of integer):integer;

вычисляющая сумму элементов массива, то в программе допустимо присваивание:

Func2:=ArraySum;

После такого присваивания инструкция

s:=Func2(x);

запишет в переменную s сумму элементов массива x (фактически будет вызвана функция ArraySum
11. Пример: интегрирование методом трапеций

Оценка интеграла, таким образом, дается формулой:

Рис. 1. Вычисление определенного интеграла методом трапеций
Составим каркас универсальной функции, вычисляющей такой интеграл:

type


TFunc = function (x: real): real;
function Example (x: real): real; 
{Та функция, от которой в последствии будем брать
begin
  интеграл}

Example:=sin(x);
end;
function Integral(a, b: real; f: TFunc): real;

<Реализация метода трапеций с использованием функции f>
begin 
{начало основной программы}


Int:=Integral(0, Pi, Example);
{Вызов функции Integral, где в качестве параметра передана 


функция Example, вычисляющая синус}

end.

Вместо Example могла бы стоять произвольная, описанная в программе функция.

Непосредственную реализацию метода трапеций выполните самостоятельно.

12. Задачи на правильное написание заголовков процедур и функций

Составление заголовков процедур и функций (по сути, выбор того, что сделать параметрами) это один из важнейших моментов написания программы. Чтобы определиться с видом заголовка требуется понять, какие данные необходимы для решения оформляемой в виде подпрограммы подзадачи (входные данные), и в каком виде лучше всего представить результат (выходные данные).

Простейший пример: требуется возвести число в квадрат. Какие данные для этого необходимы? Необходимо само число, которое будем возводить, и все. Что будет на выходе? Опять же одно число – квадрат того, что на входе. Данные на входе должны быть параметрами процедуры или функции (как правило, параметрами-значениями). Выходные данные должны записываться в параметры-переменные или в значение функции. Соответственно есть следующие варианты написания подпрограмм для решения этой задачи:

procedure NewSqr1(x: real; var x2: real);

{Входные данные передаются через параметр-значение x, результат записывается в параметр-переменную x2}

begin


x2:=x*x;

end;

function NewSqr2(x: real): real;

{Входные данные передаются через параметр-значение x, результат записывается в значение функции}

begin

NewSqr2:=x*x;

end;
procedure NewSqr3(var x: real);

{Одна и та же переменная служит для хранения входных и выходных данных}

begin


x:=x*x;

end;

Пусть есть две переменные: a, b: real; Записать в переменную b квадрат переменной a теперь можно следующими способами:

1) NewSqr1(a, b);

2) b:=NewSqr2(a);

3) b:=a;

NewSqr(b);

Некоторые студенты засовывают в заголовок описания всех переменных, которые им требуются внутри процедуры, явно путая назначение параметров и локальных переменных. Так делать нельзя! Итак:

1)
Параметрами должны быть только входные или выходные данные.

2)
Все переменные, значение которых неизвестно или несущественно в момент начала работы процедуры и значения которых не важны после окончания работы, должны быть локальными переменными.

Пример: функция для расчета факториала. Что должно быть входе? Целое число, факториал которого хотим подсчитать. Что на выходе? Значение факториала – тоже целое число. Таким образом, заголовок будет выглядеть так:

function Factorial(n: integer): integer;

При расчете потребуется переменная счетчик цикла, а также переменная, в которой будет накапливаться произведение, но все они должны быть локальными переменными.

Теперь пара слов, почему так важно правильно составлять заголовки, определяя формат данных на входе и выходе. Как уже говорилось, разработка любой сложной программы требует разбиения задачи на подзадачи, решения которых оформляются в виде процедур или функций. Далее программист сосредотачивается на решении отдельной подзадачи. Новичкам бывает трудно это сделать – начинаешь решать одну подзадачу, оказывается, для решения требуется решить другую, потом третью и т.д. Так и пытаются решить задачу целиком, что, начиная с определенного уровня сложности, просто невозможно.

Выход из этой ситуации следующий. Если, решая одну задачу, вы видите, что требуется решить подзадачу, необходимо определить, какие данные необходимы для ее решения и какие данные будут являться результатом расчетов. После этого вы можете составить заголовок процедуры или функции, решающей подзадачу. Имея заголовок, можно отложить написание тела подпрограммы на потом, а в программу вставить вызов процедуры или функции так, будто она уже написана. В результате вы решаете основную задачу, не отвлекаясь на возникающие по ходу подзадачи.

Последний вопрос. Имеется подзадача. что писать процедуру или функцию? Функция пишется тогда, когда результат имеет простой (не структурный) тип, то есть не является массивом (а также файлом, записью, классом или множеством). Кроме того, возможен вариант, когда часть выходных данных передается через параметры, а часть записывается в значение функции.

Пример 1. Опишите вещественнозначный массив и вещественную переменную, напишите заголовок подпрограммы, подсчитывающий сумму элементов массива и ее вызов с использованием описанных вами переменных.

Решение:

Задаем обязательные вопросы. Что на входе? Массив. Что на выходе? Сумма элементов – одно число. Входные данные передадим через параметр, выходные запишем в значение функции. Правильный ответ к задаче выглядит так:

const

n = 10;

var

a: array [1..n] of real;


s: real;

…

function Sum(var x: array of real): real;

…

s:=Sum(a);

Не будет ошибкой и оформление подпрограммы в виде процедуры c записью результата в параметр переменную:

procedure Sum(var x: array of real; var s: real);

…

Sum(a, s);

Пример 2. Напишите заголовок подпрограммы для нахождения суммы элементов массива, индексы которых лежат в заданном диапазоне. Напишите также вызов этой подпрограммы, описав необходимые для этого переменные.

Решение:

Задаем обязательные вопросы. Что на входе? На входе обязательно массив и требуется также задать диапазон индексов. Для этого можно сказать, начиная с какого индекса, будем суммировать и по какой (то есть задать два целых числа). Что на выходе? Если удастся подсчитать сумму, то на выходе сумма элементов (то есть число). Однако не лишним будет сделать «защиту от дурака». Что если в массиве 10 элементов, а при вызове требуется подсчитать сумму с 1-го по 20-й. Предупредить, что такое невозможно и элементов с такими индексами нет можно с помощью переменной-флага (см. параграф про переменные-флаги) – логической переменной, которая равна true, если все в порядке или false, если расчет невозможен. Таким образом, на выходе вещественное число и логическое значение.

Ответ: Правильных вариантов масса. Например, оформляем в виде функции:

const


n = 10;

var


a: array [1..n] of real;


s: real;

n1, n2: integer;

f: boolean;

…

function Sum(var x: array of real; n1, n2: integer; var flag: boolean): real;

…
s:=Sum(a, n1, n2, f);

Другой правильный вариант:

function Sum(var x: array of real; n1, n2: integer; var s: real): boolean;

…

f:=Sum(a, n1, n2, s);

Еще правильный вариант:

procedure Sum(var x: array of real; n1, n2: integer; var s: real; var f: boolean);

…

Sum(a, n1, n2, s, f);

Пример 3. Создать подпрограмму, увеличивающую все элементы массива в два раза.

Обязательные вопросы. Что на входе? Массив. Что на выходе? Такой же массив, увеличенный в 2 раза.

const


n = 10;

var


a: array [1..n] of real;

…

procedure Double(var x: array of real);

…

Double(a);

Здесь один и тот же массив x используется как входная и выходная переменная.

Еще правильный вариант:

const

n = 10;

var


a, b: array [1..n] of real;

…

procedure Double(x: array of real; var y: array of real);

…

Double(a, a);

Следующий вариант вызова не «портит» исходный массив:

Double(a, b);

Задания на составление заголовков
1. Напишите заголовки подпрограмм для решения идущих ниже задач. Запишите также вызов этих подпрограмм в основной программе, описав необходимые для этого переменные (все по аналогии с приведенными в п. 13 примерами). Если есть возможность оформить подпрограмму и как процедуру и как функцию, запишите ответ в обоих вариантах.

1) Нахождение факториала.

2) Возведение вещественного числа в целую степень.

3) Расчет среднего арифметического элементов массива.

4) Перестановка элементов массива в обратном порядке.

5) Нахождение максимального элемента числового массива.

6) Нахождение индекса максимального элемента массива.

7) Нахождение индекса максимального элемента среди последних n элементов массива.

8) Нахождение индекса максимального элемента среди первых n элементов массива.

9) Нахождение расстояния между точками в многомерном пространстве.

10) Нахождение модуля вектора.

11) Нахождение скалярного произведения.

12) Нахождение суммы двух векторов.

13) Нахождение векторного произведения двух векторов. Векторное произведение может быть рассчитано только для 3-мерных векторов. Если вектор не 3-мерный передавайте информацию о невозможности расчета через флаговую переменную.

14) Нахождение смешанного произведения.

15) Нахождение суммы двух комплексных чисел.

16) Нахождение угла между двумя векторами.

17) Вычисление гипотенузы по известным катетам.

18) В одном массиве содержатся фамилии, в другом год рождения соответствующих людей. Найти по фамилии год рождения.

19) При наличии тех же массивов, что и в предыдущей задаче узнать, сколько человек родились в заданном году.

20) При наличии тех же массивов распечатать фамилии всех людей, возраст которых лежит в заданном диапазоне.

2. Придумайте задачу, для решения которой потребуются подпрограммы со следующими заголовками:

1) procedure P1(a, b: integer);

2) procedure P2(var a, b: real);

3) procedure P3(a, b: real; var c: real);

4) function F4(x: real): real;

5) function F5(x: real): integer;

6) function F6(x, y: real): boolean;

7) function F7(x: array of real): real;

8) procedure P8(x, y: array of real; var z: array of real);

9) function F9(var s: array of string; var x: array of integer; z: integer; var flag: boolean): integer;

10) function F10(var s1, s2: array of string; var n: integer): boolean;

13. Хороший стиль при оформлении процедур и функций
Под стилем в программировании подразумеваются неформальные правила написания и оформления программ, нарушение которых допустимо, но как показал многолетний опыт, крайне нежелательно. Без серьезной на то причины делать это не следует.

Итак, при работе с процедурами и функциями следует придерживаться следующих правил:

1) Логически замкнутые куски программы следует оформлять в виде процедур и функций, даже если не предполагается выполнять заключенные в них инструкции более одного раза.

На будущее – с этого момента при решении каждой задачи хотя бы часть программы должна быть оформлена в виде процедуры или функции

2) Если значение переменной используется при расчетах внутри подпрограммы, но не требуется за ее пределами, следует делать ее локальной.

3) Предыдущее правило можно расширить: следует избегать всякого обращения внутри процедур к глобальным переменным. Весь обмен данными с главной программой должен производиться через параметры и (для функций) значения.

Следование этому правилу позволит создавать подпрограммы, работоспособность которых не зависит от остальной программы. Такие подпрограммы можно спокойно переносить из одной программы в другую. Они не начнут внезапно работать неправильно, после внесения поправок в главную программу или другие подпрограммы.

4) Все переменные-счетчики циклов, используемые в подпрограммах, должны быть локальными переменными. В принципе это правило следует из двух предыдущих, но оно на столько важно, что не мешает еще раз повторить: все переменные-счетчики циклов, используемые в подпрограммах, должны быть локальными переменными.

5) Если одним из параметров является массив, его следует сделать параметром-переменной. Это экономит память, и обращение к подпрограмме занимает меньше времени.
6) При написании заголовков в списке формальных параметров рекомендуется ставить пробелы после запятых (если есть перечисление идентификаторов через запятую), двоеточий (перед именем типа), а также после символа точка с запятой (если есть параметры разных типов). Также не помешает ставить пробел между фактическими параметрами при вызове процедуры или функции. В результате программы будут красивее смотреться и легче читаться.

7) Имена процедурам и функциям рекомендуется давать как-то связанные с теми действиями, которые они выполняют. То есть Proc1 и Func1 это, как правило, плохие имена.

